
This differs from source.cdbp in that here we use the symmetry trick (by contracting with an arbitrary symmetric
tensor) to set the symmetry on the connection.

# =============================================================================
# The connection
# =============================================================================

# --- The metric connection ---------------------------------------------------

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices.

g_{a b}::Metric.

\partial_{#}::PartialDerivative.

cderiv:=\partial_{c}{g_{a b}} - g_{a d}\Gamma^{d}_{b c}
- g_{d b}\Gamma^{d}_{a c};

Gamma:=\Gamma^{a}_{b c} -> (1/2) g^{a d} ( \partial_{b}{g_{d c}}
+ \partial_{c}{g_{b d}}
- \partial_{d}{g_{b c}} );

@substitute!(cderiv)(@(Gamma));

@distribute!(%);
@eliminate_metric!(%);
@canonicalise!(%);
@collect_terms!(%);

cderiv := 0



# =============================================================================
# One covariant derivative
# =============================================================================

@reset.

# --- Covariant differentiation -----------------------------------------------

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices.

\partial_{#}::PartialDerivative.

# --- construct the scalar v_{a} A^{a} ---------------------------------------

scalar:=v_{a} A^{a}:

# --- compute the derivative -------------------------------------------------

derivD:=D^{c}\partial_{c}{@(scalar)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{ac}A^{a}D^{c}):
@substitute!(%)(D^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{ac}D^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

# --- tidy up and display the results ----------------------------------------

@factor_out!(scalar){A^{a}}:
@factor_out!(scalar){D^{a}};

@factor_out!(derivD){A^{a}}:
@factor_out!(derivD){D^{a}};

scalar := vaAa

derivD := AaDb
(
∂bva − Γb

acvb

)

# =============================================================================
# Commutation of covariant derivatives and the Riemann tensor
# =============================================================================

@reset.

# --- covariant differentiation and the Riemann tensor ------------------------

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices.

\partial_{#}::PartialDerivative.

# --- force Gamma to be symmetric in its lower two indices --------------------

# \Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# --- construct the scalar v_{a} A^{a} ---------------------------------------

scalar:=v_{a} A^{a}:

# --- compute the covariant derivative in the direction of D^a ---------------

derivD:=D^{c}\partial_{c}{@(scalar)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(D^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}D^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):

# --- compute the covariant derivative in the direction of E^a ---------------

derivDE:=E^{c}\partial_{c}{@(derivD)}:
@distribute!(%):
@prodrule!(%):
@distribute!(%):
@substitute!(%)(E^{a}\partial_{a}{A^{b}} -> -\Gamma^{b}_{a c}A^{a}E^{c}):
@substitute!(%)(E^{a}\partial_{a}{D^{b}} -> -\Gamma^{b}_{a c}D^{a}E^{c}):
@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%);

# --- copy to derivED then swap the order of the derivatives ------------------

derivED:=@(derivDE):

@substitute!(%)(E^{a} -> F^{a}):
@substitute!(%)(D^{a} -> E^{a}):
@substitute!(%)(F^{a} -> D^{a});

# --- subtract ----------------------------------------------------------------

diff:=@(derivDE) - @(derivED);

# --- trick to impose zero torsion (symmetric connection) ---------------------

G_{a b}::Symmetric.

@substitute!(%)(\Gamma^{a}_{b c} -> G^{a} G_{b c});
@prodsort!(%):
@canonicalise!(%):
@collect_terms!(%):
@substitute!!(%)(G^{a} G_{b c} -> \Gamma^{a}_{b c});

# --- tidy up and display the results ----------------------------------------

{A^{a},D^{a},E^{a},v_{a},\Gamma^{a}_{b c}}::SortOrder.

@prodsort!(%):
@rename_dummies!(%):

@factor_out!(%){A^{a}}:
@factor_out!(%){D^{a}}:
@factor_out!(%){E^{a}}:
@factor_out!(%){v_{a}};

@print["A^a D^b E^c (v_{a;b;c}-v_{a;c;b})="~@(diff)];

derivDE := −AaDbEcΓd
ac∂bvd −AaDbEcΓd

bc∂dva + AaDbEc∂bcva + AaDbEcΓd
acΓe

dbve

+ AaDbEcΓd
aeΓe

bcvd −AaDbEc∂cΓd
abvd −AaDbEcΓd

ab∂cvd

derivED := −AaEbDcΓd
ac∂bvd −AaEbDcΓd

bc∂dva + AaEbDc∂bcva + AaEbDcΓd
acΓe

dbve

+ AaEbDcΓd
aeΓe

bcvd −AaEbDc∂cΓd
abvd −AaEbDcΓd

ab∂cvd

diff := −AaDbEcΓd
ac∂bvd −AaDbEcΓd

bc∂dva + AaDbEc∂bcva + AaDbEcΓd
acΓe

dbve + AaDbEcΓd
aeΓe

bcvd

−AaDbEc∂cΓd
abvd −AaDbEcΓd

ab∂cvd + AaEbDcΓd
ac∂bvd + AaEbDcΓd

bc∂dva −AaEbDc∂bcva

−AaEbDcΓd
acΓe

dbve −AaEbDcΓd
aeΓe

bcvd + AaEbDc∂cΓd
abvd + AaEbDcΓd

ab∂cvd

diff := −AaDbEcGdGac∂bvd −AaDbEcGdGbc∂dva + AaDbEc∂bcva + AaDbEcGdGacG
eGdbve

+ AaDbEcGdGaeG
eGbcvd −AaDbEc∂c

(
GdGab

)
vd −AaDbEcGdGab∂cvd

+ AaEbDcGdGac∂bvd + AaEbDcGdGbc∂dva −AaEbDc∂bcva −AaEbDcGdGacG
eGdbve

−AaEbDcGdGaeG
eGbcvd + AaEbDc∂c

(
GdGab

)
vd + AaEbDcGdGab∂cvd

diff := AaDbEcΓd
acΓe

bdve −AaDbEc∂cΓd
abvd −AaDbEcΓd

abΓe
cdve + AaDcEb∂cΓd

abvd

diff := AaDbEcvd

(
Γe

acΓd
be − ∂cΓd

ab − Γe
abΓd

ce + ∂bΓd
ac

)
AaDbEc (va;b;c − va;c;b) = AaDbEcvd

(
Γe

acΓd
be − ∂cΓd

ab − Γe
abΓd

ce + ∂bΓd
ac

)

# =============================================================================
# Gauss equation for 3d/4d curvatures
# =============================================================================

@reset.

# --- The Gauss equation ------------------------------------------------------

::PostDefaultRules( @@collect_terms!(%) ).
{a,b,c,d,e,f,g,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=fixed).

\nabla_{#}::Derivative.

K_{a b}::Symmetric.
g^{a}_{b}::KroneckerDelta.

# --- define the projection operator ------------------------------------------

hab:=h^{a}_{b} -> g^{a}_{b} - n^{a} n_{b}:

# --- 3-covariant derivative obtained by projection on 4-covariant derivative -

vpq:=v_{p q} -> h^{a}_{p}h^{b}_{q}\nabla_{b}{v_{a}}:

# --- compute 3-curvature by commutation of covariant derivatives -------------

vpqr:=h^{a}_{p}h^{b}_{q}h^{c}_{r} ( \nabla_{c}{v_{a b}} - \nabla_{b}{v_{a c}} ):

@substitute!(vpq)(@(hab)):
@substitute!(vpqr)(@(vpq)):

@distribute!(%):
@prodrule!(%):
@distribute!(%):
@eliminate_kr!(%):

# --- standard substitutions --------------------------------------------------

@substitute!(%)(h^{a}_{b} n^{b} -> 0):
@substitute!(%)(h^{a}_{b} n_{a} -> 0):
@substitute!(%)(\nabla_{a}{g^{b}_{c}} -> 0):
@substitute!(%)(n^{a}\nabla_{b}{v_{a}} -> -v_{a}\nabla_{b}{n^{a}}):
@substitute!(%)(v_{a}\nabla_{b}{n^{a}} -> v_{p}h^{p}_{a}\nabla_{b}{n^{a}}):
@substitute!(%)(h^{p}_{a}h^{q}_{b}\nabla_{p}{n_{q}} -> K_{a b}):
@substitute!(%)(h^{p}_{a}h^{q}_{b}\nabla_{p}{n^{b}} -> K_{a}^{q}):

# --- tidy up and display the results -----------------------------------------

{h^{a}_{b},\nabla_{a}{v_{b}}}::SortOrder.

@prodsort!(%):
@rename_dummies!(%):
@canonicalise!(%):
@factor_out!!(%){h^{a}_{b}};

vpqr := ha
ph

b
qh

c
r (∇c∇bva −∇b∇cva) + KprKq

ava −KpqKr
ava

