

First we compute the *covariant* derivatives of R^a_{bcd} in terms of its *partial* derivatives.

$$R^p_{abq;\underline{c}} B^q_p A^a A^b A^{\underline{c}} = (R^p_{abq} B^q_p)_{,\underline{c}} A^a A^b A^{\underline{c}}$$

and

$$A^a_{;b} A^b = \frac{DA^a}{Ds} = 0$$

$$B^a_{b;c} A^c = \frac{DB^a_b}{Ds} = 0$$

$$B^a_{b,c} A^c = \frac{dB^a_b}{ds} = -\Gamma^a_{dc} B^d_b A^c + \Gamma^d_{bc} B^a_d A^c$$

In the second part of this notebook we will compute the *partial* derivatives in terms of the *covariant* derivatives.

We find a curious result. For the first four derivatives the results are identical under the following changes : $\partial \rightarrow \nabla$ (on the R's only) and $\Gamma \rightarrow -\Gamma$. For the fifth derivative the terms of the form $R.d\Gamma.d^2\Gamma$ are *not* equivalent under this change. You can spot these terms, there are just two of them and they have the numerical coefficient 6.

The computations for the first four derivatives are quick – about 15 sec. But adding in just one more derivative, the fifth derivative, blows this out to about 10 mins.

```
# =====

::KeepHistory(false).
::PostDefaultRules( @@collect_terms!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices.

# we can get away with using PartialDerivative for \nabla because we are computing
# symmetrised derivatives, this makes Cadabra's job much easier

\nabla{#}::PartialDerivative.
\partial{#}::PartialDerivative.

R^{a}_{b c d}::RiemannTensor.

\delta^{a}_{b}::KroneckerDelta.

A^{a}::Depends(\partial).
B^{a u}_{b v}::Depends(\partial).
R^{a}_{b c d}::Depends(\partial).
```

```

\Gamma^{a}_{b c}::Depends(\partial).
\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# --- construct the first four symmetrised partial derivatives of the Riemann tensor -----

cderiv00:=R^{a}_{b c d} A^{b} A^{c} B^{d u}_{a v}:

cderiv01:=A^{c}\partial_{c}\{@(cderiv00)\}:
@distribute! (%):
@prodrule! (%):
@distribute! (%):
@substitute! (%)(A^{a}\partial_{a}\{A^{b}\} -> 0):
@substitute! (%)(A^{a}\partial_{a}\{B^{d u}_{c v}\} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}
-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):

cderiv02:=A^{c}\partial_{c}\{@(cderiv01)\}:
@distribute! (%):
@prodrule! (%):
@distribute! (%):
@substitute! (%)(A^{a}\partial_{a}\{A^{b}\} -> 0):
@substitute! (%)(A^{a}\partial_{a}\{B^{d u}_{c v}\} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}
-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):

cderiv03:=A^{c}\partial_{c}\{@(cderiv02)\}:
@distribute! (%):
@prodrule! (%):
@distribute! (%):
@substitute! (%)(A^{a}\partial_{a}\{A^{b}\} -> 0):
@substitute! (%)(A^{a}\partial_{a}\{B^{d u}_{c v}\} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}
-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):

```

```

cderiv04:=A^{c}\partial_{c}\{@(cderiv03)\}:
@distribute! (%):
@prodrule! (%):
@distribute! (%):
@substitute! (%)(A^{a}\partial_{a}\{A^{b}\} -> 0):
@substitute! (%)(A^{a}\partial_{a}\{B^{d u}_{c v}\} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}
-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):

cderiv05:=A^{c}\partial_{c}\{@(cderiv04)\}:
@distribute! (%):
@prodrule! (%):
@distribute! (%):
@substitute! (%)(A^{a}\partial_{a}\{A^{b}\} -> 0):
@substitute! (%)(A^{a}\partial_{a}\{B^{d u}_{c v}\} -> \Gamma^{e}_{c a}B^{d u}_{e v}A^{a}
-\Gamma^{d}_{e a}B^{e u}_{c v}A^{a}):

@distribute! (%):
@prodsort! (%):
@rename_dummies! (%):

# --- now turn to Riemann normal coordinates, defined by  $0 = \Gamma^a_{(bc,a)}$  -----
# --- temporarily replace all derivatives of \Gamma with derivatives of T -----

@substitute! (cderiv02)(\partial_{d}\{\Gamma^{a}_{b c}\} -> T_{d}^{a}_{b c}):
@substitute! (cderiv03)(\partial_{d}\{\Gamma^{a}_{b c}\} -> T_{d}^{a}_{b c}):
@substitute! (cderiv04)(\partial_{d}\{\Gamma^{a}_{b c}\} -> T_{d}^{a}_{b c}):
@substitute! (cderiv05)(\partial_{d}\{\Gamma^{a}_{b c}\} -> T_{d}^{a}_{b c}):

@substitute! (cderiv03)(\partial_{d e}\{\Gamma^{a}_{b c}\} -> T_{d e}^{a}_{b c}):
@substitute! (cderiv04)(\partial_{d e}\{\Gamma^{a}_{b c}\} -> T_{d e}^{a}_{b c}):
@substitute! (cderiv05)(\partial_{d e}\{\Gamma^{a}_{b c}\} -> T_{d e}^{a}_{b c}):

@substitute! (cderiv04)(\partial_{d e f}\{\Gamma^{a}_{b c}\} -> T_{d e f}^{a}_{b c}):
@substitute! (cderiv05)(\partial_{d e f}\{\Gamma^{a}_{b c}\} -> T_{d e f}^{a}_{b c}):

```

```

@substitute!(cderiv05)(\partial_{d e f g}\{\Gamma^a_{b c}\} -> T_{d e f g}^a_{b c}):

# --- now set \Gamma to zero -----
@substitute!!(cderiv01)(\Gamma^a_{b c} -> 0):
@substitute!!(cderiv02)(\Gamma^a_{b c} -> 0):
@substitute!!(cderiv03)(\Gamma^a_{b c} -> 0):
@substitute!!(cderiv04)(\Gamma^a_{b c} -> 0):
@substitute!!(cderiv05)(\Gamma^a_{b c} -> 0):

# --- now re-introduce the derivatives of \Gamma -----
@substitute!(cderiv02)(T_{d}^a_{b c} -> \partial_d\{\Gamma^a_{b c}\}):
@substitute!(cderiv03)(T_{d}^a_{b c} -> \partial_d\{\Gamma^a_{b c}\}):
@substitute!(cderiv04)(T_{d}^a_{b c} -> \partial_d\{\Gamma^a_{b c}\}):
@substitute!(cderiv05)(T_{d}^a_{b c} -> \partial_d\{\Gamma^a_{b c}\}):

@substitute!(cderiv03)(T_{d e}^a_{b c} -> \partial_d\{e\}\{\Gamma^a_{b c}\}):
@substitute!(cderiv04)(T_{d e}^a_{b c} -> \partial_d\{e\}\{\Gamma^a_{b c}\}):
@substitute!(cderiv05)(T_{d e}^a_{b c} -> \partial_d\{e\}\{\Gamma^a_{b c}\}):

@substitute!(cderiv04)(T_{d e f}^a_{b c} -> \partial_d\{e f\}\{\Gamma^a_{b c}\}):
@substitute!(cderiv05)(T_{d e f}^a_{b c} -> \partial_d\{e f\}\{\Gamma^a_{b c}\}):

@substitute!(cderiv05)(T_{d e f g}^a_{b c} -> \partial_d\{e f g\}\{\Gamma^a_{b c}\}):

# --- done with B, so get rid of it -----
@substitute!(cderiv01)(B^a_u_{b v} -> \delta^a_{v}\delta^u_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv02)(B^a_u_{b v} -> \delta^a_{v}\delta^u_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv03)(B^a_u_{b v} -> \delta^a_{v}\delta^u_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv04)(B^a_u_{b v} -> \delta^a_{v}\delta^u_{b}): @distribute!(%): @eliminate_kr!(%):
@substitute!(cderiv05)(B^a_u_{b v} -> \delta^a_{v}\delta^u_{b}): @distribute!(%): @eliminate_kr!(%):

# --- tidy up and display the results -----
tmp01:=@(cderiv01): @canonicalise!(%): @factor_out!!(%){A^a}:
tmp02:=@(cderiv02): @canonicalise!(%): @factor_out!!(%){A^a}:
tmp03:=@(cderiv03): @canonicalise!(%): @factor_out!!(%){A^a}:
tmp04:=@(cderiv04): @canonicalise!(%): @factor_out!!(%){A^a}:
tmp05:=@(cderiv05): @canonicalise!(%): @factor_out!!(%){A^a}:

```

```

@substitute!(tmp01)(\partial_{p}{R^{a}_{b d c}} -> - \partial_{p}{R^{a}_{b c d}}):
@substitute!(tmp02)(\partial_{p q}{R^{a}_{b d c}} -> - \partial_{p q}{R^{a}_{b c d}}):
@substitute!(tmp03)(\partial_{p q r}{R^{a}_{b d c}} -> - \partial_{p q r}{R^{a}_{b c d}}):
@substitute!(tmp04)(\partial_{p q r s}{R^{a}_{b d c}} -> - \partial_{p q r s}{R^{a}_{b c d}}):
@substitute!(tmp05)(\partial_{p q r s t}{R^{a}_{b d c}} -> - \partial_{p q r s t}{R^{a}_{b c d}}):

@print["\cderivA="~@(tmp01)~"?"];
@print["\cderivB="~@(tmp02)~"?"];
@print["\cderivC="~@(tmp03)~"?"];
@print["\cderivD="~@(tmp04)~"?"];
@print["\cderivE="~@(tmp05)~"?"];

```

The first five symmetrised *covariant* derivatives of R^a_{bcd} in a Riemann normal coordinate frame.

$$A^a A^b A^c R^u_{bcv;a} = A^a A^b A^c \partial_a R^u_{bcv}$$

$$A^a A^b A^c A^d R^u_{cdv;ab} = A^a A^b A^c A^d (\partial_{ab} R^u_{cdv} + R_{vabg} \partial_c \Gamma^u_{dg} - R^u_{abg} \partial_c \Gamma^g_{vd})$$

$$A^a A^b A^c A^d A^e R^u_{dev;abc} = A^a A^b A^c A^d A^e (3 \partial_a R_{v b c h} \partial_d \Gamma^u_{eh} - 3 \partial_a R^u_{b c h} \partial_d \Gamma^h_{ve} + \partial_{abc} R^u_{dev} + R_{vabh} \partial_{cd} \Gamma^u_{eh} - R^u_{abh} \partial_{cd} \Gamma^h_{ve})$$

$$\begin{aligned} A^a A^b A^c A^d A^e A^f R^u_{efv;abcd} = A^a A^b A^c A^d A^e A^f (6 \partial_a \Gamma^u_{bi} \partial_{cd} R_{vefi} + 4 \partial_a R_{v b c i} \partial_{de} \Gamma^u_{fi} - 6 \partial_a \Gamma^i_{vb} \partial_{cd} R^u_{efi} - 4 \partial_a R^u_{b c i} \partial_{de} \Gamma^i_{vf} + \partial_{abcd} R^u_{efv} + 3 R_{vabi} \partial_c \Gamma^u_{dj} \partial_e \Gamma^j_{fi} \\ + 6 R^a_{ibj} \partial_c \Gamma^u_{di} \partial_e \Gamma^j_{vf} + R_{vabi} \partial_{cde} \Gamma^u_{fi} + 3 R^u_{abi} \partial_c \Gamma^j_{vd} \partial_e \Gamma^i_{fj} - R^u_{abi} \partial_{cde} \Gamma^i_{vf}) \end{aligned}$$

$$\begin{aligned} A^a A^b A^c A^d A^e A^f A^g R^u_{fgv;abcd} = A^a A^b A^c A^d A^e A^f A^g (15 \partial_a R_{v b c j} \partial_d \Gamma^u_{ek} \partial_f \Gamma^k_{gj} + 30 \partial_a R^b_{j c k} \partial_d \Gamma^u_{ej} \partial_f \Gamma^k_{vg} + 10 \partial_{ab} R_{v c d j} \partial_{ef} \Gamma^u_{gj} + 10 \partial_a \Gamma^u_{bj} \partial_{cde} R_{v f g j} \\ + 5 \partial_a R_{v b c j} \partial_{def} \Gamma^u_{gj} + 15 \partial_a R^u_{b c j} \partial_d \Gamma^k_{ve} \partial_f \Gamma^j_{gk} - 10 \partial_{ab} R^u_{c d j} \partial_{ef} \Gamma^j_{vg} - 10 \partial_a \Gamma^j_{vb} \partial_{cde} R^u_{f g j} - 5 \partial_a R^u_{b c j} \partial_{def} \Gamma^j_{vg} + \partial_{abcde} R^u_{f g v} \\ + 4 R_{v a b j} \partial_c \Gamma^k_{dj} \partial_{ef} \Gamma^u_{gk} + 6 R_{v a b j} \partial_c \Gamma^u_{dk} \partial_{ef} \Gamma^k_{gj} + 10 R^a_{j b k} \partial_c \Gamma^j_{vd} \partial_{ef} \Gamma^u_{gk} + 10 R^a_{j b k} \partial_c \Gamma^u_{dj} \partial_{ef} \Gamma^k_{vg} + R_{v a b j} \partial_{c d e f} \Gamma^u_{g j} \\ + 6 R^u_{a b j} \partial_c \Gamma^k_{vd} \partial_{ef} \Gamma^j_{gk} + 4 R^u_{a b j} \partial_c \Gamma^j_{dk} \partial_{ef} \Gamma^k_{vg} - R^u_{a b j} \partial_{c d e f} \Gamma^j_{vg}) \end{aligned}$$

Now we compute the *partial* derivatives of R^a_{bcd} in terms of its *covariant* derivatives.

We do this by swapping term across the equal sign in the previous result followed by substitutions of lower order partial derivatives from previous results. Huh?

A typical result of the previous calculations will be of the form

$$R^a_{(bcd;pq)} = R^a_{(bcd;pq)} + Q^a_{bcdpq}$$

where Q contains lower order partial derivatives of R (and various derivatives of the connections). We rewrite this as

$$R^a_{(bcd;pq)} = R^a_{(bcd;pq)} - Q^a_{bcdpq}$$

The lower order partial derivatives of R that appear on the right hand side can be eliminated by substitutions from previous results.

```
# --- now re-arrange equations to give partial derivs in terms of covariant derivs -----

tmp01:= - @(cderiv01):
tmp02:= - @(cderiv02):
tmp03:= - @(cderiv03):
tmp04:= - @(cderiv04):
tmp05:= - @(cderiv05):

# --- replace highest order \partial with \nabla -----

@substitute!(tmp01)(\partial_{p}\{R^{a}_{b c d}\} -> - \nabla_{p}\{R^{a}_{b c d}\}):
@substitute!(tmp02)(\partial_{p q}\{R^{a}_{b c d}\} -> - \nabla_{p q}\{R^{a}_{b c d}\}):
@substitute!(tmp03)(\partial_{p q r}\{R^{a}_{b c d}\} -> - \nabla_{p q r}\{R^{a}_{b c d}\}):
@substitute!(tmp04)(\partial_{p q r s}\{R^{a}_{b c d}\} -> - \nabla_{p q r s}\{R^{a}_{b c d}\}):
@substitute!(tmp05)(\partial_{p q r s t}\{R^{a}_{b c d}\} -> - \nabla_{p q r s t}\{R^{a}_{b c d}\}):

# --- use previous results to eliminate lower order \partial's -----

@substitute!(tmp02)(A^{a} A^{b} A^{c}\partial_{c}\{R^{u}_{a b v}\} -> @(tmp01)):
@substitute!(tmp03)(A^{a} A^{b} A^{c}\partial_{c}\{R^{u}_{a b v}\} -> @(tmp01)):
@substitute!(tmp04)(A^{a} A^{b} A^{c}\partial_{c}\{R^{u}_{a b v}\} -> @(tmp01)):
@substitute!(tmp05)(A^{a} A^{b} A^{c}\partial_{c}\{R^{u}_{a b v}\} -> @(tmp01)):

@substitute!(tmp03)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}\{R^{u}_{a b v}\} -> @(tmp02)):
@substitute!(tmp04)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}\{R^{u}_{a b v}\} -> @(tmp02)):
@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d}\partial_{c d}\{R^{u}_{a b v}\} -> @(tmp02)):

@substitute!(tmp04)(A^{a} A^{b} A^{c} A^{d} A^{e}\partial_{c d e}\{R^{u}_{a b v}\} -> @(tmp03)):
@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d} A^{e}\partial_{c d e}\{R^{u}_{a b v}\} -> @(tmp03)):
```

```

@substitute!(tmp05)(A^{a} A^{b} A^{c} A^{d} A^{e} A^{f}\partial_{c d e f}\{R^{u}_{a b v}\} -> @(tmp04)):

@distribute!(tmp01):
@distribute!(tmp02):
@distribute!(tmp03):
@distribute!(tmp04):
@distribute!(tmp05):

# --- tidy up and display the results -----
pderiv01:=@(tmp01): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv02:=@(tmp02): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv03:=@(tmp03): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv04:=@(tmp04): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:
pderiv05:=@(tmp05): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%): @factor_out!!(%){A^{a}}:

@substitute!(pderiv01)(\nabla_{p}\{R^{a}_{b d c}\} -> - \nabla_{p}\{R^{a}_{b c d}\}):
@substitute!(pderiv02)(\nabla_{p q}\{R^{a}_{b d c}\} -> - \nabla_{p q}\{R^{a}_{b c d}\}):
@substitute!(pderiv03)(\nabla_{p q r}\{R^{a}_{b d c}\} -> - \nabla_{p q r}\{R^{a}_{b c d}\}):
@substitute!(pderiv04)(\nabla_{p q r s}\{R^{a}_{b d c}\} -> - \nabla_{p q r s}\{R^{a}_{b c d}\}):
@substitute!(pderiv05)(\nabla_{p q r s t}\{R^{a}_{b d c}\} -> - \nabla_{p q r s t}\{R^{a}_{b c d}\}):

@print["\pderivA="~@(pderiv01)~"?"];
@print["\pderivB="~@(pderiv02)~"?"];
@print["\pderivC="~@(pderiv03)~"?"];
@print["\pderivD="~@(pderiv04)~"?"];
@print["\pderivE="~@(pderiv05)~"?"];

```


The first five symmetrised *partial* derivatives of R^a_{bcd} in a Riemann normal coordinate frame.

$$A^a A^b A^c R^u_{bcv,a} = A^a A^b A^c \nabla_a R^u_{bcv}$$

$$A^a A^b A^c A^d R^u_{cdv,ab} = A^a A^b A^c A^d (\nabla_{ab} R^u_{cdv} - R_{vabe} \partial_c \Gamma^u_{de} + R^u_{abe} \partial_c \Gamma^e_{vd})$$

$$A^a A^b A^c A^d A^e R^u_{dev,abc} = A^a A^b A^c A^d A^e (-3 \nabla_a R_{vbcf} \partial_d \Gamma^u_{ef} + 3 \nabla_a R^u_{bcf} \partial_d \Gamma^f_{ve} + \nabla_{abc} R^u_{dev} - R_{vabf} \partial_{cd} \Gamma^u_{ef} + R^u_{abf} \partial_{cd} \Gamma^f_{ve})$$

$$\begin{aligned} A^a A^b A^c A^d A^e A^f R^u_{efv,abcd} = A^a A^b A^c A^d A^e A^f & (-6 \nabla_{ab} R_{vcdg} \partial_e \Gamma^u_{fg} + 3 R_{vabg} \partial_c \Gamma^u_{dh} \partial_e \Gamma^h_{fg} + 6 R^a_{gbh} \partial_c \Gamma^u_{dg} \partial_e \Gamma^h_{vf} - 4 \nabla_a R_{vbcg} \partial_{de} \Gamma^u_{fg} + 6 \nabla_{ab} R^u_{cdg} \partial_e \Gamma^g_{vf} \\ & + 3 R^u_{abg} \partial_c \Gamma^h_{vd} \partial_e \Gamma^g_{fh} + 4 \nabla_a R^u_{bcg} \partial_{de} \Gamma^g_{vf} + \nabla_{abcd} R^u_{efv} - R_{vabg} \partial_{cde} \Gamma^u_{fg} + R^u_{abg} \partial_{cde} \Gamma^g_{vf}) \end{aligned}$$

$$\begin{aligned} A^a A^b A^c A^d A^e A^f A^g R^u_{fgv,abcd} = A^a A^b A^c A^d A^e A^f A^g & (15 \nabla_a R_{vbch} \partial_d \Gamma^u_{ei} \partial_f \Gamma^i_{gh} + 30 \nabla_a R^b_{hci} \partial_d \Gamma^u_{eh} \partial_f \Gamma^i_{vg} - 10 \nabla_{ab} R_{vcdh} \partial_{ef} \Gamma^u_{gh} + 6 R_{vabh} \partial_c \Gamma^i_{dh} \partial_{ef} \Gamma^u_{gi} \\ & + 10 R^a_{hbi} \partial_c \Gamma^h_{vd} \partial_{ef} \Gamma^u_{gi} - 10 \nabla_{abc} R_{vdeh} \partial_f \Gamma^u_{gh} + 4 R_{vabh} \partial_c \Gamma^u_{di} \partial_{ef} \Gamma^i_{gh} + 10 R^a_{hbi} \partial_c \Gamma^u_{dh} \partial_{ef} \Gamma^i_{vg} - 5 \nabla_a R_{vbch} \partial_{def} \Gamma^u_{gh} \\ & + 15 \nabla_a R^u_{bch} \partial_d \Gamma^i_{ve} \partial_f \Gamma^h_{gi} + 10 \nabla_{ab} R^u_{cdh} \partial_{ef} \Gamma^h_{vg} + 6 R^u_{abh} \partial_c \Gamma^h_{di} \partial_{ef} \Gamma^i_{vg} + 10 \nabla_{abc} R^u_{deh} \partial_f \Gamma^h_{vg} + 4 R^u_{abh} \partial_c \Gamma^i_{vd} \partial_{ef} \Gamma^h_{gi} \\ & + 5 \nabla_a R^u_{bch} \partial_{def} \Gamma^h_{vg} + \nabla_{abcde} R^u_{fgv} - R_{vabh} \partial_{cdef} \Gamma^u_{gh} + R^u_{abh} \partial_{cdef} \Gamma^h_{vg}) \end{aligned}$$